

Formalization of the pumping lemma for context-free languages

Marcus V. M. Ramos

Centro de Informática, UFPE, Recife, Brazil

Ruy J. G. B. de Queiroz

Centro de Informática, UFPE, Recife, Brazil

Nelma Moreira

*Departamento de Ciência de Computadores, Faculdade de Ciências, Universidade do Porto,
Porto, Portugal*

José Carlos Bacelar Almeida

Departamento de Informática, Universidade do Minho, Braga, Portugal

Abstract

Context-free languages (CFLs) are highly important in computer language processing technology as well as in formal language theory. The Pumping Lemma is a property that is valid for all context-free languages, and is used to show the existence of non context-free languages. This paper presents a formalization, using the Coq proof assistant, of the Pumping Lemma for context-free languages.

Keywords: Context-free languages, Pumping Lemma, formalization, formal mathematics, proof assistant, interactive proof systems, Coq.

1. Introduction

The formalization of context-free language theory is key to the certification of compilers and programs, as well as to the development of new languages and tools for certified programming.

We aim to formalize a substantial part of context-free language theory in the Coq proof assistant, making it possible to reason about it in a fully checked environment, with all the related advantages. Initially, however, the focus has

Email addresses: mvmr@cin.ufpe.br (Marcus V. M. Ramos), ruy@cin.ufpe.br (Ruy J. G. B. de Queiroz), nam@dcc.fc.up.pt (Nelma Moreira), jba@di.uminho.pt (José Carlos Bacelar Almeida)

been restricted to context-free grammars and associated results. Pushdown automata and their relation to context-free grammars are not considered at this point.

The work, that started with the formalization of closure properties for context-free grammars [11], evolved later into the formalization of context-free grammar simplification [12] and then into the Chomsky normalization of context-free grammars [13]. Formalization of simplification enabled the formalization of the Chomsky normalization, which in turn enabled the present formalization of the Pumping Lemma.

In order to follow this paper, the reader is required to have basic knowledge of Coq and of context-free language theory. For the beginner, the recommended starting point for Coq is the book by Bertot and Castéran [5]. Background on context-free language theory can be found in [16] or [14], among others. Previous results, which were used in the formalization of the Pumping Lemma, will not be discussed here and can be retrieved from the above references.

The statement and applications of the Pumping Lemma for CFLs (or Pumping Lemma for short) are presented in Section 2. A typical informal proof, which served as the basis for the present formalization, is described in Section 3. Section 4 introduces results obtained previously and which are required for this work. The formalization is then described in Section 5, where the definitions and auxiliary lemmas used are discussed in some detail, as well as the Pumping Lemma itself. Section 6 discusses related work by various other researchers and final conclusions are presented in Section 7.

As far as the authors are aware of, this is the first formalization of the Pumping Lemma for context-free languages in any proof assistant. All the definitions and proof scripts discussed in this paper were written in plain Coq and are available for download at:

<https://github.com/mvmramos/pumping>

2. Statement and Application

A language is a set of words defined over an alphabet. A context-free grammar is a grammar whose rules have the form $X \rightarrow \beta$, where X is a non-terminal symbol and β is a sequence (possibly empty) of terminal and non-terminal symbols. A context-free language is a language that is generated by some context-free grammar. The Pumping Lemma is a property that is verified for all CFLs and was stated and proved for the first time by Bar-Hillel, Perles and Shamir in 1961 [1].

The Pumping Lemma does not characterize the CFLs, however, since it is also verified by some non CFLs. It states that, for every context-free language and for every sentence of such a language that has a certain minimum length, it is possible to obtain an infinite number of new sentences that must also belong to the language. This minimum length depends only on the language defined. In other words (let \mathcal{L} be defined over alphabet Σ):

$$\begin{aligned}
& \forall \mathcal{L}, (\text{cfl } \mathcal{L}) \rightarrow \exists n \mid \\
& \forall \alpha, (\alpha \in \mathcal{L}) \wedge (|\alpha| \geq n) \rightarrow \\
& \exists u, v, w, x, y \in \Sigma^* \mid (\alpha = uvwxy) \wedge (|vx| \geq 1) \wedge (|vwx| \leq n) \wedge \\
& \forall i, uv^iwx^iy \in \mathcal{L}
\end{aligned}$$

A typical use of the Pumping Lemma is to show that a certain language is not context-free by using the contrapositive of the statement of the lemma. The proof proceeds by contraposition: the language is assumed to be context-free, and this leads to a contradiction from which one concludes that the language in question cannot be context-free.

As an example, consider the language $\mathcal{L} = \{a^i b^i c^i \mid i \geq 1\}$. This language is defined over the alphabet $\{a, b, c\}$ and includes sentences such as $abc, aabbcc, aaabbbccc$.

Should \mathcal{L} be context-free, then the Pumping Lemma should hold for it. Consider n to be the constant of the Pumping Lemma and let's choose the sentence $\alpha = a^n b^n c^n$. Clearly, $\alpha \in \mathcal{L}$ and $|\alpha| = 3 * n \geq n$. Thus, $\alpha = uvwxy$ such that $|vx| \geq 1$, $|vwx| \leq n$ and $uv^iwx^iy \in \mathcal{L}, i \geq 0$.

However, it is easy to observe that, due to its length limitation, the sentence vwx should contain only one or two different symbols (namely, vwx should belong to either a^*, b^*, c^*, a^*b^* or b^*c^*). This implies that the repetition of v and x in uv^iwx^iy should increase (or decrease) the number of at most two different symbols while keeping the number of the third symbol unchanged. As a result, the new sentence cannot belong to the language and this proves that the initial hypothesis cannot be true. Thus, \mathcal{L} is not a context-free language.

3. Informal Proof

In short, the Pumping Lemma derives from the fact that the number of non-terminal symbols in the grammar G that generates \mathcal{L} is finite. The classical proof considers that G is in the Chomsky Normal Form (a form in which the rules of the grammar have at most two symbols in the right-hand side), which means that derivation trees have the simpler form of binary trees. Then, if the sentence has a certain minimum length, the frontier of the derivation tree should have two or more instances of the same non-terminal symbol in some path that starts in the root of this tree. Finally, the context-free character of G guarantees that the subtrees related to these duplicated non-terminal symbols can be cut and pasted in such a way that an infinite number of new derivation trees are obtained, each of which is related to a new sentence of the language.

The proof comprises the following steps (more details can be found in [16] or [14]):

1. Since \mathcal{L} is declared to be a context-free language (predicate `cfl`), then there exists a context-free grammar G such that $L(G) = \mathcal{L}$;
2. Obtain G' such that G' is in Chomsky Normal Form and $L(G') = L(G)$;

3. Take n as 2^k , where k is the number of non-terminal symbols in G' ;
4. Choose α such that $\alpha \in \mathcal{L}$ and $|\alpha| \geq n$;
5. Obtain a derivation tree t that represents the derivation of α in G' ;
6. Take a path that starts in the root of t and whose length is the height of t plus 1 (maximum length);
7. Then, the height of t should be greater or equal than $k + 1$;
8. This means that the selected path has at least $k + 2$ symbols, being at least $k + 1$ non-terminals and one (the last) a terminal symbol;
9. Since G' has only k non-terminal symbols, this means that this path has at least one non-terminal symbol that appears at least two times in it;
10. Name the duplicated symbols n_1 and n_2 ($n_1 = n_2$) and the corresponding subtrees t_1 and t_2 (note that t_2 is a subtree of t_1 and t_1 is a subtree of t);
11. It is then possible to prove that the height of t_1 is greater than or equal to 2, and less than or equal to 2^k ;
12. Also, that the height of t_2 is greater than or equal to 1 and less than or equal to 2^{k-1} ;
13. This implies that the frontier of t can be split into five parts: u, v, w, x, y , where w is the frontier of t_2 and $vw x$ is the frontier of t_1 ;
14. As a consequence of the heights of the corresponding subtrees, it can be shown that $|vx| \geq 1$ and $|vwx| \leq n$;
15. If t_1 is removed from t , and t_2 is inserted in its place, then we have a new tree t^0 that represents the derivation of string $uv^0wx^0y = uwy$;
16. If, instead, t_1 is inserted in the place where t_2 lies originally, then we have a tree t^2 that represents the derivation of string uv^2wx^2y ;
17. Repetition of the previous step generates all trees t^i that represent the derivation of the string $uv^iwx^iy, \forall i \geq 2$.

4. Background

Formalization in the Coq proof assistant requires the formalization of the Chomsky Normal Form (CNF) for context-free grammars. This, in turn, demands the formalization of context-free grammar simplification (useless and inaccessible symbol elimination and unit and empty rules elimination). For details on how these have been accomplished, please refer to [12] and [13].

The Chomsky Normal Form (CNF) theorem asserts

$$\begin{aligned} \forall G = (V, \Sigma, P, S), \exists G' = (V', \Sigma, P', S') \mid \\ L(G) = L(G') \wedge \\ \forall (\alpha \rightarrow_{G'} \beta) \in P', (\beta \in \Sigma) \vee (\beta \in N \cdot N) \end{aligned}$$

That is, every context-free grammar can be converted to an equivalent one whose rules have only one terminal symbol or two non-terminal symbols in the right-hand side. Naturally, this is valid only if G does not generate the empty string. If this is the case, then the grammar that has this format, plus

a single rule $S' \rightarrow_{G'} \epsilon$, is also considered to be in the Chomsky Normal Form, and generates the original language, including the empty string. It can also be assured that in either case the start symbol of G' does not appear on the right-hand side of any rule of G' .

The CNF theorem has been stated in our formalization as:

```
Theorem g_cnf:
  ∀ g: cfg non_terminal terminal,
  (produces_empty g ∨ ∼ produces_empty g) ∧
  (produces_non_empty g ∨ ∼ produces_non_empty g) →
  ∃ g': cfg non_terminal' terminal,
  g_equiv g' g ∧
  (is_cnf g' ∨ is_cnf_with_empty_rule g').
```

Context-free grammars are represented by record `cfg` in a way that resembles the definition:

```
Record cfg (non_terminal terminal:Type): Type := {
  start_symbol: non_terminal;
  rules: non_terminal → list (non_terminal + terminal) → Prop;
  rules_finite:
    ∃ n: nat,
    ∃ ntl: list non_terminal,
    ∃ t1: list terminal,
    rules_finite_def start_symbol rules n ntl t1}.
```

The predicate `rules_finite_def` assures that the set of rules of the grammar is finite by proving that the length of right-hand side of every rule is equal or less than a given value, and also that both left and right-hand side of the rules are built from finite sets of, respectively, non-terminal and terminal symbols (represented here by lists).

The predicate `produces g s` asserts that context-free grammar `g` produces the list of terminals `s` as a sentence of the language. It is based on the more fundamental notion of derivation, present in the whole formalization and defined as:

```
Inductive derives (g: cfg): sf → sf → Prop :=
| derives_refl:
  ∀ s: sf,
  derives g s s
| derives_step:
  ∀ s1 s2 s3: sf,
  ∀ left: non_terminal,
  ∀ right: sf,
  derives g s1 (s2 ++inl left :: s3) →
  rules g left right →
  derives g s1 (s2 ++right ++s3).
```

The predicates used in theorem `g_cnf_final` above assert that:

- a grammar produces the empty string:

Definition produces_empty
 (g: cfg non_terminal terminal): Prop:=
 produces g [].

- a grammar produces a non-empty string:

Definition produces_non_empty
 (g: cfg non_terminal terminal): Prop:=
 $\exists s: \text{sentence}, \text{produces } g \ s \wedge s \neq []$.

- two grammars are equivalent:

Definition g_equiv
 (non_terminal non_terminal' terminal: Type)
 (g1: cfg non_terminal terminal)
 (g2: cfg non_terminal' terminal): Prop:=
 $\forall s: \text{sentence},$
 produces g1 s \leftrightarrow produces g2 s.

- a rule is in the Chomsky Normal Form:

Definition is_cnf_rule
 (left: non_terminal) (right: sf): Prop:=
 $(\exists s1 \ s2: \text{non_terminal}, \text{right} = [\text{inl } s1; \text{inl } s2]) \vee$
 $(\exists t: \text{terminal}, \text{right} = [\text{inr } t]).$

- a grammar is in the Chomsky Normal Form:

Definition is_cnf
 (g: cfg non_terminal terminal): Prop:=
 $\forall \text{left}: \text{non_terminal},$
 $\forall \text{right}: \text{sf},$
 rules g left right \rightarrow is_cnf_rule left right.

- a grammar is in the Chomsky Normal Form and has a single empty rule with the start symbol in the left-hand side:

Definition is_cnf_with_empty_rule
 (g: cfg non_terminal terminal): Prop:=
 $\forall \text{left}: \text{non_terminal},$
 $\forall \text{right}: \text{sf},$
 rules g left right \rightarrow
 $(\text{left} = (\text{start_symbol } g) \wedge \text{right} = []) \vee$
 is_cnf_rule left right.

5. Formalization

The formalization follows closely the steps described in Section 3. The Pumping Lemma has been stated as follows:

```

Lemma pumping_lemma:
  ∀ l: lang terminal,
  (contains_empty l ∨ ~ contains_empty l) ∧
  (contains_non_empty l ∨ ~ contains_non_empty l) →
  cfl l →
  ∃ n: nat,
  ∀ s: sentence,
  l s →
  length s ≥ n →
  ∃ u v w x y: sentence,
  s = u ++v ++w ++x ++y ∧
  length (v ++x) ≥ 1 ∧
  length (v ++w ++x) ≤ n ∧
  ∀ i: nat, l (u ++(iter v i) ++w ++(iter x i) ++y).

```

A language is defined as a function that maps a sentence (a list of terminal symbols) to a proposition (**Prop**):

```

Definition lang (terminal Type):= list terminal → Prop.

```

Two languages are equal if they have the same sentences:

```

Definition lang_eq (l k: lang) :=
  ∀ w, l w ↔ k w.

```

Finally, a language is context-free if it is generated by some context-free grammar:

```

Definition cfl (terminal: Type) (l: lang terminal): Prop:=
  ∃ non_terminal: Type,
  ∃ g: cfg non_terminal terminal,
  lang_eq l (lang_of_g g).

```

where `lang_of_g` represents the language generated by grammar `g`:

```

Definition lang_of_g (g: cfg non_terminal terminal): lang :=
  fun w: sentence ⇒ produces g w.

```

Predicates `contains_empty` and `contains_non_empty` are language counterparts of the previously presented grammar predicates `produces_empty` and `produces_non_empty`, respectively. Application `iter l i` on a list `l` and a natural `i` yields list `li`.

Initially, the type `btree` (for binary trees) has been defined with the objective of representing derivation trees for strings generated by context-free grammars in Chomsky Normal Form:

```

Inductive btree (non_terminal terminal: Type): Type:=
| bnode_1: non_terminal → terminal → btree
| bnode_2: non_terminal → btree → btree → btree.

```

The constructors of `btree` relate to the two possible forms that the rules of a CNF grammar can assume (namely with one terminal symbol or two non-terminal symbols in the right-hand side).

The proof of the Pumping Lemma starts by finding a grammar G that generates the input language L (this is a direct consequence of the predicate `is_cfl` and corresponds to step 1 of Section 3). Next, we obtain a CNF grammar G' that is equivalent to G (step 2), using previous results. Then, G is substituted for G' and the value for n is defined as 2^k (step 3) where k is the length of the list of non-terminals of G' (which in turn is obtained from the predicate `rules_finite`).

Lemma `derives_g_cnf_equiv_btree` is then applied in order to obtain a `btree` t that represents the derivation of α in G' (step 5). This lemma is general enough in order to accept that the input grammar might either be a CNF grammar, or a CNF grammar with an empty rule. If this is the case, then we have to ensure that $\alpha \neq \epsilon$, which is true since by assumption $|\alpha| \geq 2^k$. The proof of `derives_g_cnf_equiv_btree` is reasonably long and uses induction on the number of derivation steps in G' in order to generate α :

```

Lemma derives_g_cnf_equiv_btree:
  ∀ g: cfg non_terminal' terminal,
  ∀ n: non_terminal',
  ∀ s: sentence,
  s ≠ [] →
  (is_cnf g ∨ is_cnf_with_empty_rule g) →
  start_symbol_not_in_rhs g →
  derives g [inl n] (map term_lift' s) →
  ∃ t: btree non_terminal' terminal,
  btree_cnf g t ∧
  broot t = n ∧
  bfrontier t = s.

```

The next step is to obtain a path (a sequence of non-terminal symbols ended by a terminal symbol) that has maximum length, that is, whose length is equal to the height of t plus 1 (steps 6 and 7). This is accomplished by means of the definition `bpath` and the lemma `btree_ex_bpath`:

```

Inductive bpath (bt: btree): sf → Prop :=
| bp_1: ∀ n: non_terminal,
  ∀ t: terminal,
  bt = (bnode_1 n t) → bpath bt [inl n; inr t]
| bp_l: ∀ n: non_terminal,
  ∀ bt1 bt2: btree,
  ∀ p1: sf,
  bt = bnode_2 n bt1 bt2 → bpath bt1 p1 → bpath bt ((inl n) :: p1)
| bp_r: ∀ n: non_terminal,
  ∀ bt1 bt2: btree,
  ∀ p2: sf,
  bt = bnode_2 n bt1 bt2 → bpath bt2 p2 → bpath bt ((inl n) :: p2).

```

```

Lemma btree_ex_bpath:
  ∀ bt: btree,
  ∀ ntl: list non_terminal,
  bheight bt ≥ length ntl + 1 →

```



```

bnts bt ntl →
∃ z: sf,
bpath bt z ∧
length z = bheight bt + 1 ∧
∃ u r: sf,
∃ t: terminal,
z = u ++r ++[inr t] ∧
length u ≥ 0 ∧
length r = length ntl + 1 ∧
(∀ s: symbol, In s (u ++r) → In s (map inl ntl)).

```

The length of this path (which is $\geq k + 2$) allows one to infer that it must contain at least one non-terminal symbol that appears at least twice in it (steps 8, 9 and 10). This result comes from the application of the lemma `pigeon` which represents a list version of the well-known pigeonhole principle:

```

Lemma pigeon:
∀ A: Type,
∀ x y: list A,
(∀ e: A, In e x → In e y) →
length x = length y + 1 →
∃ d: A,
∃ x1 x2 x3: list A,
x = x1 ++[d] ++x2 ++[d] ++x3.

```

Since a path is not unique in a tree, it is necessary to use some other representation that can describe this path uniquely, which is done by the predicate `bcode` and the lemma `bpath_ex_bcode`:

```

Inductive bcode (bt: btree): list bool → Prop :=
| bcode_0: ∀ n: non_terminal,
  ∀ t: terminal,
  bt = (bnode_1 n t) → bcode bt []
| bcode_1: ∀ n: non_terminal,
  ∀ bt1 bt2: btree,
  ∀ c1: list bool,
  bt = bnode_2 n bt1 bt2 → bcode bt1 c1 → bcode bt (false :: c1)
| bcode_2: ∀ n: non_terminal,
  ∀ bt1 bt2: btree,
  ∀ c2: list bool,
  bt = bnode_2 n bt1 bt2 → bcode bt2 c2 → bcode bt (true :: c2).

```

The predicate `bcode` uses a sequence of boolean values (`false`, `true`) to respectively select the left or right subtrees in a tree, and thus define a path in it.

```

Lemma bpath_ex_bcode:
∀ t: btree,
∀ p: sf,
bpath t p →
∃ c: list bool,
bcode t c ∧

```

`bpath_bcode t p c`.

The predicate `bpath_bcode` merely ensures that `bcode c` is valid for `bpath p` in tree `t`. Once the path has been identified with a repeated non-terminal symbol, and a corresponding `bcode` has been assigned to it, lemma `bcode_split` is applied twice in order to obtain the two subtrees t_1 and t_2 that are associated respectively to the first and second repeated non-terminals of t . This lemma, which is key in the formalization, has a statement with a number of hypothesis and conclusions which give many useful informations on the newly identified subtree. Among them, its height and its frontier (this one embedded in the definition `btree_decompose`):

```
Lemma bcode_split:
  ∀ t: btree,
  ∀ p1 p2: sf,
  ∀ c: list bool,
  bpath_bcode t (p1 ++p2) c →
  length p1 > 0 →
  length p2 > 1 →
  bheight t = length p1 + length p2 - 1 →
  ∃ c1 c2: list bool,
  c = c1 ++c2 ∧
  length c1 = length p1 ∧
  ∃ t2: btree,
  ∃ x y: sentence,
  bpath_bcode t2 p2 c2 ∧
  btree_decompose t c1 = Some (x, t2, y) ∧
  bheight t2 = length p2 - 1.
```

Function `btree_decompose` takes as arguments a tree and a sequence of boolean values, and returns a triple consisting of the subtree located in this position and the two sentences to the left and right of it. It is used to enable reasoning on the frontiers of the subtrees obtained before.

From this information it is then possible to extract most of the results needed to prove the goal (steps 11, 12, 13 and 14), except for the pumping condition. This has been obtained by an auxiliary lemma `pumping_aux`, which takes as hypothesis the fact that a tree t_1 (with frontier vwx) has a subtree t_2 (with frontier w), both with the same roots, and asserts the existence of an infinite number of new trees obtained by repeated substitution of t_2 by t_1 or simply t_1 by t_2 , with respectively frontiers $v^iwx^i, i \geq 1$ and w , or simply $v^iwx^i, i \geq 0$:

```
Lemma pumping_aux:
  ∀ g: cfg _ _,
  ∀ t1 t2: btree (non_terminal' non_terminal terminal) _,
  ∀ n: _,
  ∀ c1 c2: list bool,
  ∀ v x: sentence,
  btree_decompose t1 c1 = Some (v, t2, x) →
  btree_cnf g t1 →
  broot t1 = n →
```

```

bcode t1 (c1 ++c2) →
c1 ≠ [] →
broot t2 = n →
bcode t2 c2 →
(∀ i: nat,
 ∃ t': btree _ _,
 btree_cnf g t' ∧
 broot t' = n ∧
 btree_decompose t' (iter c1 i) = Some (iter v i, t2, iter x i) ∧
 bcode t' (iter c1 i ++c2) ∧
 get_nt_btree (iter c1 i) t' = Some n).

```

The proof continues by showing that each of these new trees can be combined with tree t obtained before, thus representing strings uv^iwx^iy , $i \geq 0$ as necessary (steps 15 and 16).

Finally, it must be proved that each of these trees is related to a derivation in G' , which is accomplished by lemma `btree_equiv_produces_g_cnf`, the dual version of lemma `derives_g_cnf_equiv_btree` (step 17).

The Pumping Lemma has some 400 lines of Coq script, which adds to auxiliary lemmas and an extensive library of lemmas on binary trees and on the relation of binary trees to Chomsky Normal Form grammars. The whole approach is constructive, except for the proof of the `pigeon` lemma, which was formalized with classical logic extensions.

6. Related Work

Context-free language theory formalization is a relatively new area of research, with some results already obtained with a diversity of proof assistants, including Coq, HOL4 and Agda. Most of the effort started in 2010 and have been devoted to the certification and validation of parser generators. Examples of this are the works of Koprowski and Binsztok (using Coq, [10]), Ridge (using HOL4, [15]), Jourdan, Pottier and Leroy (using Coq, [9]) and, more recently, Firsov and Uustalu (in Agda, [7]).

On the more theoretical side, on which the present work should be considered, Norrish and Barthwal published on general context-free language theory formalization using the using HOL4 proof assistant [2, 3, 4], including the existence of normal forms for grammars, pushdown automata and closure properties. Recently, Firsov and Uustalu proved the existence of a Chomsky Normal Form grammar for every general context-free grammar, using the Agda proof assistant [8].

A special case of the Pumping Lemma, namely the Pumping Lemma for regular languages, is included in a comprehensive work on the formalization of regular languages [6] using SSRreflect, an extension of Coq.

7. Conclusions

The formalization of the Pumping Lemma for context-free languages represents the culmination of an effort that started with closure properties for

context-free grammars [11] and continued with simplification for context-free grammars [12] and the Chomsky Normal Form [13]. The whole formalization has 20.000+ lines of Coq script and was developed over a period of two years.

The Pumping Lemma is a significant result in language theory in general and this is, as far as the authors are aware of, the first formalization ever of it, 54 years after it was stated and proved for the first time. It has to be seen against the backdrop of the important and well sought after goal of formalizing fundamental results in language theory, as well as formalizing mathematics in general.

The libraries developed to support this formalization will hopefully play an equally important role, as they include general results on context-free language theory that can be used or adapted to prove other results. The whole work can serve to different purposes, including the continued formalization of language theory and the teaching of formal languages, formalization and Coq itself.

Bibliography

References

- [1] Bar-Hillel, Y., Perles, M. A., Shamir, E., 1961. On formal properties of simple phrase structure grammars. *Zeitschrift für Phonetik, Sprachwissenschaft und Kommunikationsforschung* (14), 143–172.
- [2] Barthwal, A., Norrish, M., August 2010. A formalisation of the normal forms of context-free grammars in HOL4. In: Dawar, A., Veith, H. (Eds.), *Computer Science Logic, 24th International Workshop, CSL 2010, 19th Annual Conference of the EACSL, Brno, Czech Republic, August 23–27, 2010. Proceedings*. Vol. 6247 of *Lecture Notes in Computer Science*. Springer, pp. 95–109.
- [3] Barthwal, A., Norrish, M., 2010. Mechanisation of PDA and grammar equivalence for context-free languages. In: Dawar, A., de Queiroz, R. J. G. B. (Eds.), *Logic, Language, Information and Computation, 17th International Workshop, WoLLIC 2010*. Vol. 6188 of *Lecture Notes in Computer Science*. pp. 125–135.
- [4] Barthwal, A., Norrish, M., 2014. A mechanisation of some context-free language theory in HOL4. *Journal of Computer and System Sciences (WoLLIC 2010 Special Issue, A. Dawar and R. de Queiroz, eds.)* 80 (2), 346 – 362. URL <http://www.sciencedirect.com/science/article/pii/S0022000013000925>
- [5] Bertot, Y., Castéran, P., 2004. *Interactive Theorem Proving and Program Development*. Springer.
- [6] Doczkal, C., Kaiser, J.-O., Smolka, G., 2013. A constructive theory of regular languages in Coq. In: Gonthier, G., Norrish, M. (Eds.), *Certified Programs and Proofs*. Vol. 8307 of *Lecture Notes in Computer Science*. Springer

International Publishing, pp. 82–97.
URL http://dx.doi.org/10.1007/978-3-319-03545-1_6

- [7] Firsov, D., Uustalu, T., 2014. Certified {CYK} parsing of context-free languages. *Journal of Logical and Algebraic Methods in Programming* 83 (5–6), 459 – 468, the 24th Nordic Workshop on Programming Theory (NWPT 2012).
URL <http://www.sciencedirect.com/science/article/pii/S2352220814000601>
- [8] Firsov, D., Uustalu, T., 2015. Certified normalization of context-free grammars. In: *Proceedings of the 2015 Conference on Certified Programs and Proofs. CPP '15*. ACM, New York, NY, USA, pp. 167–174.
URL <http://doi.acm.org/10.1145/2676724.2693177>
- [9] Jourdan, J.-H., Pottier, F., Leroy, X., 2012. Validating LR(1) parsers. In: *Proceedings of the 21st European Conference on Programming Languages and Systems. ESOP'12*. Springer-Verlag, Berlin, Heidelberg, pp. 397–416.
URL http://dx.doi.org/10.1007/978-3-642-28869-2_20
- [10] Koprowski, A., Binsztok, H., 2010. TRX: A formally verified parser interpreter. In: *Proceedings of the 19th European Conference on Programming Languages and Systems. ESOP'10*. Springer-Verlag, Berlin, Heidelberg, pp. 345–365.
URL http://dx.doi.org/10.1007/978-3-642-11957-6_19
- [11] Ramos, M. V. M., de Queiroz, R. J. G. B., Jun. 2015. Formalization of closure properties for context-free grammars. *ArXiv e-prints*.
URL <http://arxiv.org/abs/1506.03428>
- [12] Ramos, M. V. M., de Queiroz, R. J. G. B., 2015. Formalization of simplification for context-free grammars. In: *Preliminary Proceedings of the 10th Workshop on Logical and Semantic Frameworks, with Applications. LSFA'15*.
- [13] Ramos, M. V. M., de Queiroz, R. J. G. B., Moreira, N., Almeida, J. C. B., 2016. Formalization of context-free language theory. [Submitted for publication].
- [14] Ramos, M. V. M., Neto, J. J., Vega, I. S., 2009. *Linguagens Formais: Teoria Modelagem e Implementação*. Bookman.
- [15] Ridge, T., 2011. Simple, functional, sound and complete parsing for all context-free grammars. In: *CPP*. pp. 103–118.
- [16] Sudkamp, T. A., 2006. *Languages and Machines*, 3rd Edition. Addison-Wesley.